# Maintainer: Daurnimator <daurnimator@archlinux.org>
# Contributor: Willem Mulder <14mRh4X0r@gmail.com>

pkgname=firefox-extension-passff
pkgver=1.15.0
pkgrel=1
pkgdesc="zx2c4 pass manager addon for firefox"
url="https://github.com/passff/passff"
license=('GPL2')
arch=('any')
groups=('firefox-addons')
replaces=('firefox-passff')
depends=('firefox' 'passff-host')
source=("$pkgname-$pkgver.xpi::https://github.com/passff/passff/releases/download/${pkgver%.0}/passff.xpi")
sha512sums=('990aa520e57e223ae43c86cd326e826fab1eeb16b16c534e36eb9c1637da528218f8eb39e8f79b0310bd8329eadac4889ce20649155e708817e8fbfac1068bd9')
noextract=("$pkgname-$pkgver.xpi")

package() {
    install -Dm644 "$pkgname-$pkgver.xpi" \
        "${pkgdir}/usr/lib/firefox/browser/extensions/passff@invicem.pro.xpi"
}
